
Abstract This paper presents a PVS development of relevant results of the theory
of rings. The PVS theory includes complete proofs of the three classical isomor-
phism theorems for rings, and characterizations of principal, prime and maximal
ideals. Algebraic concepts and properties are specified and formalized as generally
as possible allowing in this manner their application to other algebraic struc-
tures. The development provides the required elements to formalize important
algebraic theorems. In particular, the paper presents the formalization of the gen-
eral algebraic-theoretical version of the Chinese Remainder Theorem (CRT) for
the theory of rings, as given in abstract algebra textbooks, proved as a consequence
of the First Isomorphism Theorem. Also, the PVS theory includes a formalization
of the number-theoretical version of CRT for the structure of integers, which is
the version of CRT found in formalizations. CRT for integers is obtained as a
consequence of the general version of CRT for the theory of rings.

Keywords PVS · Ring theory · Isomorphism theorems · Principal ideals · Prime
ideals · Maximal ideals · Chinese Remainder Theorem for Rings

1 Introduction

This work presents formalizations of the three isomorphism theorems for rings,
characterizations of principal, prime and maximal ideals and a formal proof of the
algebraic version of the Chinese Remainder Theorem developed in the Prototype
Verification System (PVS). These developments belong to the PVS theory rings

that specify and formalize notions and properties of rings as given in classical refer-
ences on abstract algebra such as Hungerford’s [20] and, Dummit and Foote’s [12]
textbooks. Indeed, Hungerford’s textbook is the main reference for this formaliza-
tion and the results about homomorphisms, prime, maximal and principal ideals

follow the same presentation of this book, whereas the Chinese Remainder Theo-
rem follows the approach in Dummit and Foote’s textbook. Other references (such
as Herstein’s, Artin’s and Jacobson’s textbooks [19,2,22]) were employed mainly
to consult different pen-and-paper proofs and to compare with related work.

Formalizing the theory of rings in PVS is mainly motivated by two reasons:

– Firstly, ring theory has a wide range of applications in several fields of knowl-
edge. For example, important areas such as combinatorics, algebraic cryptog-
raphy and coding theory apply finite commutative rings [5]. Regarding coding
theory, commutative rings with identity and polynomials over such structures
have an important role to describe redundant codes according to Lidl and
Niederreiter [26]. Also, ring theory forms the basis for algebraic geometry,
which has applications in engineering systems, statistics, modeling of biolog-
ical processes and computer algebra [30]. Other relevant application of ring
theory is in the analysis of images. According to Suárez et. al. [35], if the
images are considered as a matrix whose elements belong to the ring Zn, prop-
erties about restoration and segmentation of images and detection of contours
can be competently described. In particular, Suárez et. al. [35] showed that the
segmentation of digital images becomes more efficiently automated by applying
the Zn to obtain index of similarity between images. Thus, a formalization of
the main results of ring theory would make possible the formal verification of
more complex theories involving rings in their scope.

– Secondly, fully formalizing the theory of rings contributes to the enrichment
of libraries of mathematics in PVS. An important and well-known library for
PVS is the NASA PVS Library4 (nasalib) that contains many theories in sev-
eral subjects, such as analysis [7], topology [24], term rewriting systems [13],
among others. In particular, nasalib contains the theory algebra [6], where
basic concepts and properties about groups, rings and fields were formalized.
However, the contents of the theory algebra about rings was restricted to def-
initions and just basic results obtained from such definitions. The first steps
to enrich the theory algebra gave rise to the theory rings, and initially in-
cluded elaborated formalizations such as the Binomial Theorem for rings, a
result establishing that every finite integral domain with cardinality greater
than one is a field (i.e., commutative division ring or skew field) and the First
Isomorphism Theorem, as shortly reported by three of the authors in [34]. Ho-
momorphisms and cosets were separately defined in the particular context of
rings and ideals in [34] and just for groups and normal subgroups in [6]. To
the best of our knowledge, the only formalizations on rings in PVS are in the
theories algebra of nasalib and in the theory rings.

PVS is a proof assistant with a logic core that is based on classical higher-
order logic, and that embeds the power of subtyping and dependent typing in
its logic. Specifications in PVS are functional and the PVS proof engine follows a
Gentzen’s sequent calculus style. The higher-order features of PVS are adequate for
specifying theorems about properties and relations, such as morphisms, quotients,
ideals, among others, that appear naturally in the study of algebraic structures
such as groups, rings, fields, etc. As a proof assistant based on classical logic, PVS
includes the middle excluded rule, and the PVS core logic includes, in its prelude,

4 Available at https://shemesh.larc.nasa.gov/fm/ftp/larc/PVS-library/

the axiom of choice. The development of the theory rings as well as the whole
theory algebra don’t include any axioms other than those provided as part of the
core logic of PVS. The semantics of PVS, and the relations between PVS proof
commands and deductive rules can be consulted respectively in [28] and [3]. This
paper discusses all necessary details about PVS required for understanding the
description of the theory rings.

The main contributions of this work are listed below.

(i) Formalization of preliminary concepts on algebraic structures in a manner that
allows their use in general contexts involving structures different from rings.
All the sub-theories of the theory rings in [34] related to homomorphisms,
cosets and quotient rings as well as the formalization of the First Isomorphism
Theorem, were restructured. In this work, homomorphisms are specified for
magmas (as used in Bourbaki’s Éléments de Mathématique), which are denoted
with the name groupoids in the theory algebra, and cosets for arbitrary sets.
Besides making the specification of the three isomorphism theorems clearer and
proof obligations easier, the new specifications of homomorphisms and cosets
could be applied to other structures than rings and groups.

(ii) Formalization of the three isomorphism theorems for rings that are classical
landmarks in abstract algebra.

(iii) Formalization of definitions and properties about principal, maximal and prime
ideals. Correctness of an algebraic description of the elements of a principal
ideal is proved. Furthermore, alternative characterizations of prime and maxi-
mal ideals are provided: prime ideals in a commutative ring are characterized
by an algebraic condition on their elements, and prime and maximal ideals in
commutative rings with identity are characterized by quotient rings.

(iv) A formal proof of the general algebraic version of the Chinese Remainder The-
orem (CRT) for rings obtained as a consequence of the First Isomorphism
Theorem for rings is presented. Also, to illustrate the usability of the develop-
ment, the number-theoretical version of the CRT for integers was formalized as
an application of the general case. The CRT for integers depends on a specifi-
cation that required formalizations of specific properties about the ring Zn. As
pointed out before, since (finite) commutative rings in general, and specially
the ring Zn, have relevant applications, the development of formalizations for
(finite) rings plays an important role as the basis of formalizations for other
theories.

The paper is organized as follows: Section 2 presents a theoretical overview
about the isomorphism theorems and theorems on ideals that were formalized,
pointing out the required concepts and results. Section 3 details the interesting
elements of the specification and formalization of the required basic definitions,
algebraic structures and their properties. Section 4 discusses the interesting as-
pects about the formalization of the three isomorphism theorems for rings, while
Section 5 the formalization of principal, maximal and prime ideals. Section 6 in-
troduces the formalization of the general algebraic version of CRT for rings proved
as a consequence of the First Isomorphism Theorem and, Section 7 discusses the
formalization of the standard number-theoretical version of that theorem for in-
tegers. After that, Section 8 discusses related work and, finally, Section 9 con-
cludes and brings suggestions for future works. The theory rings is available at
https://github.com/nasa/pvslib (following the master branch for PVS 7.1) as

https://github.com/nasa/pvslib

part of the nasalib, inside the theory algebra. The development is also locally
available at http://ayala.mat.unb.br/publications.html, where instructions
for checking the theory are provided in the README top rings file.

2 Isomorphism theorems for rings and properties of ideals

The first subsection discusses the necessary background about the three Isomor-
phism theorems for rings and the second subsection all the required background
related with the properties formalized for principal, maximal and prime ideals.

2.1 Isomorphism theorems for rings

A ring is an algebraic structure described by a quadruple (R,+R, ∗R, zeroR), where
R is a nonempty set that is closed under the binary operations addition and
multiplication, denoted respectively as +R and ∗R, over R and contains a neutral
additive element, namely the constant zeroR. Furthermore, a ring is an Abelian
group restricted to (R,+R, zeroR) and the associative law holds for the operation
∗R as well as the right and left distributive laws of ∗R over +R. If x ∗R y =
y ∗R x for all x, y ∈ R then (R,+R, ∗R, zeroR) is said to be a commutative ring.
Furthermore, if R contains an element oneR such that oneR ∗R x = x ∗R oneR = x
then (R,+R, ∗R, zeroR, oneR) is called a ring with identity. More information about
rings and other structures in abstract algebra, including those crucial for this
formalization such as ideals, morphisms, kernel, cosets, quotients, etc. can be found
in the previous cited books ([20,19,12,2]).

Basically, the isomorphism theorems for rings are results about homomor-
phisms between structures that are, in almost all cases, quotient rings. In the
next paragraphs, the isomorphism theorems for rings will be enunciated in order
to highlight the concepts and the algebraic structures involved in the formalization
of such propositions. In the following, for simplicity, a ring (R,+R, ∗R, zeroR) will
be denoted by R and the subscripts of the operations will be omitted.

The First Isomorphism Theorem states that an arbitrary homomorphism from
a ring R to a ring S induces an isomorphism between two specific rings, as pre-
sented below.

Theorem 1 (First Isomorphism Theorem) If R and S are rings and ϕ : R→
S is a homomorphism then there is an isomorphism of rings from R/ker(ϕ) to the
image of ϕ, where ker(ϕ) denotes the kernel of the ring homomorphism ϕ.

The Second Isomorphism Theorem deals with an isomorphism between two
quotient rings involving intersection and sum of ideals, whereas the Third Isomor-
phism Theorem is about an isomorphism between a quotient of quotient rings and
another quotient ring.

Theorem 2 (Second Isomorphism Theorem) Let H and I be a subring and
an ideal in a ring R, respectively. There is an isomorphism between the rings
H/(H ∩ I) and (H + I)/I, where H + I = {x ∈ R | x = h+ i, h ∈ H and i ∈ I}.

http://ayala.mat.unb.br/publications.html

Theorem 3 (Third Isomorphism Theorem) Let J and I be ideals in a ring
R. If J ⊂ I then I/J is an ideal in R/J and there is an isomorphism of rings from
(R/J)/(I/J) to R/I.

Theorems 1, 2 and 3 require some concepts and structures that deserve special
mention, namely: ring homomorphisms, kernel of a homomorphism, ideals in a
ring and quotient rings R/I. A full formalization of the properties of these objects
was necessary in order to obtain the formalization of the isomorphism theorems for
rings in PVS, as presented in the sub-theories ring 1st isomorphism theorem and
ring 2nd 3rd isomorphism theorems (see Figure 1). The sub-theories of the the-
ory algebra that were relevant for the formalization of the isomorphism theorems
are just groupoid def and ring.

ring def

ring

algebra
rings

ring cosets lemmas

homomorphism def

ring homomorphism def

ring 1st isomorphism theorem

ring 2nd 3rd isomorphism theorems

cosets def

ring ideal

ring basic properties ring ideal def

groupoid def

product coset def

ring homomorphism lemmas quotient rings

Fig. 1 Hierarchy of the sub-theories for the three isomorphism theorems for rings:
ring 1st isomorphism theorem and ring 2nd 3rd isomorphism theorems

2.2 Principal, prime and maximal ideals

Ideals are notable structures in ring theory since they play for rings a similar role
as normal subgroups play for groups; for example, it is possible to characterize
ideals as a kernel of a ring homomorphism. Formalizations of properties about
principal, prime and maximal ideals are considered in this work.

Let R be a ring and a an element of R. Consider the family of all ideals of R
which contain a. The intersection of the ideals in this family is called a principal
ideal generated by a and denoted as (a). Essentially, (a) is the smallest ideal of
R containing a. Theorem 4 provides an interesting characterization of principal
ideals.

Theorem 4 (Principal Ideals) Let R be a ring and a an element of R:

(i) the principal ideal (a) corresponds to the set {r∗a+a∗s+n·a+
∑m

i=1 ri∗a∗si |
r, s, ri, si ∈ R;m ∈ N \ {0};n ∈ Z}, where n · a denotes n summands of a if
n ≥ 0, and n summands of −a if n < 0;

(ii) if R is a commutative ring then (a) = {r ∗ a+ n · a | r ∈ R;n ∈ Z};
(iii) if R is a commutative ring and has an identity then (a) = {r ∗ a} = {a ∗ r},

where r ∈ R.

An ideal P of a ring R is called a prime ideal if P 6= R and for any ideals A,B
in R, one has that A ∗ B ⊂ P implies A ⊂ P or B ⊂ P , where A ∗ B = {x ∈ R |
x = a ∗ b, a ∈ A and b ∈ B}. Theorem 5 establishes a characterization of prime
ideals for commutative rings.

Theorem 5 (Prime Ideals for Commutative Rings) Let R be a ring, not
necessarily commutative. If P is an ideal in R such that P 6= R and for all a, b ∈ R
it holds that

a ∗ b ∈ P ⇒ a ∈ P or b ∈ P (1)

then P is prime. Reciprocally, if P is a prime ideal in R and R is a commutative
ring then P satisfies condition (1).

On the other hand, Theorem 6 provides a characterization of prime ideals with
identity by a quotient ring.

Theorem 6 (Prime Ideals for Rings with Identity) Let R be a commutative
ring with identity one 6= zero. An ideal P in R is prime if and only if the quotient
ring R/P is an integral domain.

The concept of prime ideals is related with the notion of prime numbers in the
set of integers Z. In fact, it is well-known that nZ = {n ∗ z | n ∈ N; z ∈ Z} is
an ideal in Z. Notice that if nZ is a prime ideal then n 6= 1 (since nZ 6= Z) and,
according to Theorem 5, whenever a ∗ b is an element of nZ, one has that a or b is
a member of nZ. In other words, this means that if n divides a ∗ b then n divides
a or b, whence nZ is a prime ideal if n is prime.

Lastly, an ideal M in a ring R is said to be maximal if M 6= R and for any
ideal N in R such that M ⊂ N ⊂ R either N = M , or N = R. Maximal ideals in
commutative rings, under the condition stated in Theorem 7, are prime ideals.

Theorem 7 (Maximal Ideals in Commutative Rings) If R is a commutative
ring such that R ∗R = R and M is a maximal ideal in R then M is a prime ideal.

Theorem 8 establishes a connection between maximal ideals and the nature of
some quotient rings.

Theorem 8 (Maximal Ideals and Quotient Rings) Consider an ideal M in
a ring R with identity:

(i) if R is a commutative ring and M is a maximal ideal then the quotient ring
R/M is a field;

(ii) if the quotient ring R/M is a division ring (skew field) then M is a maximal
ideal.

Results of this subsection were formalized in sub-theories of rings depicted
in Figure 2. Theorem 4 was formalized in the sub-theories ring principal ideal

(items (i) and (ii)) and ring with id one generator (item (iii)), Theorems 5, 6, 7
and 8 were formalized in the sub-theories ring prime ideal, ring with one prime

ideal, ring maximal ideal and ring with one maximal ideal, respectively. The
sub-theories of the theory algebra that are imported for the formalization of
the properties on ideals discussed here are ring with one, integral domain def,
field def, ring, and ring nz closed def.

ring

ring nz closed def

ring with one

field def

integral domain def

ring with one ideal

ring basic properties

rings ring principal idealring with one maximal ideal

ring principal ideal def

ring with one prime ideal ring maximal ideal

ring with id one generator

ring prime ideal

ring with one basic properties

integral domain with one def

quotient rings with one

ring maximal ideal def

ring prime ideal def

ring one generator

cosets def

ring ideal

quotient rings

ring ideal def

algebra

ring def
ring with one def

Fig. 2 Hierarchy of the sub-theories related with principal, prime and maximal ideals

3 Formalization of morphisms and basic agebraic structures

In this section, the PVS sub-theories of rings given in Figure 1 and related with
specification and formalization of morphisms and basic algebraic notions such as
cosets, ideals, kernel and quotient are presented.

3.1 Formalization of homomorphisms

The concept of ring homomorphism is the core of the theory for isomorphism theo-
rems (see Figure 1). The required definitions and properties about homomorphisms
were formalized in the sub-theories homomorphisms def, ring homomorphisms def

and ring homomorphism lemmas(Figure 1 shows these sub-theories in the hierar-
chy). Particularly, the subtheory homomorphisms def presents an interesting con-
tribution, once the definition of homomorphism was formalized for closed sets
under a specific operation. This is a more general PVS formalization about ho-
momorphisms of algebraic structures than the one given in the theory algebra

of nasalib. In the subtheory ring homomorphisms def, the concept of homomor-
phism was extended in order to make feasible its use in a context of rings and the
subtheory ring homomorphism lemmas brings the formalization of classical results
involving ring homomorphisms.

3.1.1 Subtheory homomorphisms def

Before the development of the theory rings, started in [34], the only specifica-
tion in PVS of homomorphism for algebraic structures was available in subtheory
homomorphisms of the theory algebra [6] (see Specification 1).

Specification 1 Homomorphism for groups - subtheory nasalib/algebra@homomorphisms

homomorphism ?(G1: group[T1,*,one1], G2: group[T2,o,one2],
phi: [(G1) -> (G2)]): bool =

(FORALL (a,b: (G1)): phi(a*b) = phi(a) o phi(b))

The parameters of a group are a nonempty type T, a binary operator *: [T,

T-> T], and a constant of type T. Thus, in the specification above G1 and G2 are
groups of respective types group[T1,*,one1] and group[T2,o,one2] over the re-
spective types T1 and T2, and with their associated binary operators and constants.
In this specification, a function phi of type [(G1) -> (G2)], that is a function
from G1 to G2 is an homomorphism, i.e., satisfies the predicate homomorphism?,
whenever FORALL (a,b: (G1)): phi(a*b) = phi(a) o phi(b) holds. From this,
the property phi(one1) = one2 can be inferred. Observe that this specification is
specialized for groups. Two situations arise from this particular choice to formalize
the notion of homomorphism:

(i) obviously, such definition makes sense only in a context dealing with groups;
(ii) if such specification is used to describe a homomorphism ϕ : G→ H, as proof

obligations generated from the type checking, it is necessary to verify that G
and H are groups. In other words, one must check that G and H are closed
under specific operations and the associative law holds, a neutral element for
such operations belongs to G and H and every element of G and H has an
inverse.

In the new approach, as shown in the code in Specification 2, a more general
definition of homomorphism was specified in the subtheory homomorphisms def

for magmas.

Specification 2 Specification of homomorphism in the subtheory homomorphisms def

R: VAR (groupoid ?[T,s])
S: VAR (groupoid ?[U,p])

homomorphism ?(R, S)(phi: [(R) -> (S)]): bool =
FORALL(a,b: (R)): phi(s(a,b)) = p(phi(a),phi(b))

This specification requires that R and S, declared as variables, which are spec-
ified as VAR in PVS of respective types (groupoid?[T,s]) and (groupoid?[U,p]),
be magmas. A magma (G, ∗) is just a nonempty set G closed under the bi-
nary operation ∗. Consequently, the specification of homomorphism developed
in homomorphisms def can be used in contexts involving monoids, semigroups,
groups, rings, fields or any other one that deal with a structure containing a
nonempty set satisfying the closure property for some operation. In this way, this
choice results more adaptable and malleable than the one specialized for groups
and available in nasalib/algebra@homomorphisms. Furthermore, the proof obli-
gations generated by the new specification of homomorphism are easier to check,
since verifying whether a structure is a magma has a much lower cost than check-
ing whether the structure is a group. For structures as rings with two operators
over the same type (typically, for multiplication and addition) this specification
of homomorphism is used twice to guarantee that the (homomorphism) property
holds for both operators.

3.1.2 Subtheory ring homomorphisms def

Since the definition of a ring homormorphism must consider homomorphisms re-
lated with each one of the two operations over a ring R, to facilitate the formal-
ization of the properties about ring homomorphisms, the specification of the pred-
icate R homomorphism?(R 1,R 2)(phi) was provided (see Specification 3). This
curryfied predicate has as first parameters two sets R 1 and R 2, whose elements
have arbitrary types T 1 and T 2, respectively. In addition, the binary opera-
tions s 1 and p 1 are closed over R 1, such that groupoid?[T 1,s 1](R 1) and
groupoid?[T 1,p 1](R 1), whereas s 2 and p 2 are also closed over R 2, such that
groupoid?[T 2,s 2](R 2) and groupoid?[T 2,p 2](R 2) . This is done in sub-
theory ring homomorphisms def. Since R homomorphism?(R1, R2)(phi) makes use
of the definition homomorphism?(R1, R2)(phi) specified in homomorphisms def, one
must ensure that R 1 and R 2 are magmas under the operations mentioned above.

Specification 3 Specification of R homomorphism?(R1, R2)(phi)

R_homomorphism ?(R1, R2)(phi: [(R1) -> (R2)]): bool =
groupoid ?[T1,s1](R1) AND groupoid ?[T1,p1](R1) AND
groupoid ?[T2,s2](R2) AND groupoid ?[T2,p2](R2) AND
homomorphism ?[T1,s1,T2,s2](R1, R2)( phi) AND
homomorphism ?[T1,p1,T2,p2](R1, R2)( phi)

Injective, surjective and bijective homomorphisms are also formalized in the
subtheory ring homomorphisms def, as presented in Specification 4.

Specification 4 Specification of injective, surjective and bijective homomorphisms

R_monomorphism ?(R1,R2)( phi: [(R1) -> (R2)]): bool =
injective ?(phi) AND R_homomorphism ?(R1,R2)(phi)

R_epimorphism ?(R1,R2)(phi: [(R1) -> (R2)]): bool =
surjective ?(phi) AND R_homomorphism ?(R1,R2)(phi)

R_isomorphism ?(R1,R2)(phi: [(R1) -> (R2)]): bool =
R_monomorphism ?(R1,R2)( phi) AND R_epimorphism ?(R1,R2)(phi)

3.1.3 Subtheory ring homomorphism lemmas

This subtheory mainly aims to formalize results about ring homomorphisms; within
its scope, rings R 1 and R 2 are considered and, since rings are magmas, a sim-
pler characterization of ring homomorphisms obtained from the specification of
ring homomorphism, R homomorphism?(R1, R2)(phi), was formalized, as presented
in Specification 5.

Specification 5 Characterization of a ring homomorphism

R_homo_equiv: LEMMA
FORALL(phi :[(R1)->(R2)]): R_homomorphism ?(R1,R2)( phi) IFF
FORALL(x,y:(R1)): phi(s1(x,y)) = s2(phi(x),phi(y)) AND

phi(p1(x,y)) = p2(phi(x),phi(y))

Furthermore, classical results were formally proved in this subtheory, among
them:

– if ϕ : R→ S is a ring homomorphism and x is an element of R then ϕ(zeroR) =
zeroS and ϕ(inv(x)) = inv(ϕ(x)), where inv(x) denotes the inverse of the
element x according to the operation +R;

– the image of a ring homomorphism is a ring;
– the kernel of a homormorphism is an ideal;
– the characterization of an injective homomorphism ϕ based on kerϕ.

3.2 Formalization of basic algebraic structures

Quotient rings are the main structures involved in the isomorphism theorems,
besides that, they are involved in results that provide alternative characterizations
of prime and maximal ideals. However, the construction of quotient rings depends
on other objects such as ideals and cosets. The next subsections present these
algebraic structures and formalizations of their properties.

3.2.1 Subtheory cosets def

Let T be a non-interpreted type that is any abstract type, and let G, H and I
be sets provided of a binary operation +, whose elements have the same non-
interpreted type T. Furthermore, let g be an arbitrary element of type T. In the
subtheory cosets def, the sets g+H, H + g and H + I were specified as +(g, H),
+(H, g) and sum(H, I), respectively, as illustrated in Specification 6.

Specification 6 The sets g + H, g + H and H + I

+(g,H): set[T] = {t:T | EXISTS (h:(H)): t = g+h} ;

+(H,g): set[T] = {t:T | EXISTS (h:(H)): t = h+g} ;

sum(H,I): set[T] = {t:T | EXISTS (h:(H), k:(I)): t = h+k}

Since the elements of H and I have arbitrary types, the specification estab-
lished as sum(H, I) can be used in the formalization of the Second Isomorphism

Theorem, where H is a subring and I is an ideal of a ring R, as well as in other
contexts involving the sum or any other binary operation of two arbitrary struc-
tures containing elements of the same type. For example, the hypothesis R∗R = R
in Theorem 7 was specified as sum[T,*](R,R) = R. The sets +(g, H) and +(H, g)
were used to specify the notion of cosets in a general way (see Specification 7). Also,
generators of left and right cosets were defined as lc gen(G,H) and rc gen(G,H).
The use of the operator choose for specifying left and right coset generators re-
quires proving proof obligations generated by typechecking: non of the sets {a: T |

G(a) AND A = a + H} and {a: T | G(a) AND A = H + a}. PVS proves automat-
ically both these proof obligations by expanding the definitions of left coset?

and right coset? and by Skolemization.

Specification 7 Definition of cosets for arbitrary sets G and H

left_coset?(G,H)(A:set[T]): bool = (EXISTS(a:(G)): A = a+H)

right_coset ?(G,H)(A:set[T]): bool = (EXISTS(a:(G)): A = H+a)

coset?(G,H)(A:set[T]): bool =
left_coset?(G,H)(A) AND right_coset?(G,H)(A)

lc_gen(G,H)(A:left_coset(G,H)) : T =
choose ({a: T | G(a) AND A = a + H})

rc_gen(G,H)(A:right_coset(G,H)) : T =
choose ({a: T | G(a) AND A = H + a})

In nasalib/algebra@cosets, one can find another specification of left and
right cosets and cosets (see Specification 8). However, the formalization of cosets
in the subtheory cosets def has advantages in relation to that one since it is
restricted to groups. Besides that, since cosets are defined over arbitrary types in
cosets def, in general, there are no proof obligations generated by type checking,
whereas one must verify whether determined structure is a group, when the code
in Specification 8 is considered.

Specification 8 Cosets in the context of groups in nasalib/algebra@cosets

left_coset(G:group ,H:subgroup(G))(a:(G)):
{s: set[T] | subset ?(s,G)} = a*H

3.2.2 Ideals

Ideals have a similar role for ring theory as normal subgroups have for group theory
and a quotient ring R/I is well-defined when one considers the left cosets of a ring
R modulo an ideal I in R under specific operations, which shows that ideals are
important structures to establish quotient rings. The subtheory ring ideal def

brings the specification of the definition of ideals as shown in Specification 9.

Specification 9 Definition of ideals

R: VAR (ring?)
I: VAR set[T]

left_swallow ?(I,R): bool =
FORALL (r:(R), x:(I)): member(r * x,I)

right_swallow ?(I,R): bool =
FORALL (r:(R), x:(I)): member(x * r,I)

left_ideal?(I,R): bool =
subring ?(I,R) AND left_swallow ?(I,R)

right_ideal ?(I,R): bool =
subring ?(I,R) AND right_swallow ?(I,R)

ideal?(I,R): bool = left_ideal ?(I,R) AND right_ideal?(I,R)

Several properties of ideals were formalized in the subtheory ring ideal, among
them those given in Specification 10.

Specification 10 Properties of ideals

R: VAR ring

ideal_equiv: LEMMA
ideal?(I,R) IFF
(nonempty ?(I) AND subset ?(I,R) AND
FORALL (x,y:(I), r:(R)): member(x - y,I) AND

member(x*r,I) AND member(r*x,I))

ideal_transitive: LEMMA
subring ?(H,R) AND ideal ?(I,R) AND subset ?(I, H)

IMPLIES ideal?(I,H)

intersection_subring_ideal: LEMMA
subring ?(H,R) AND ideal ?(I,R)

IMPLIES ideal?( intersection(H,I),H)

The Lemma ideal equiv provides a simpler characterization of ideals that
is applied in the proofs of lemmas such as intersection subring ideal and
ideal transitive as well as in other proofs where it is required to show that
some structure is an ideal. Both the lemmas intersection subring ideal and
ideal transitive are crucial for checking proof obligations generated from the
Second Isomorphism Theorem, where it is required a quotient ring involving the
intersection between a subring H and an ideal I in a ring R. The variable R is de-
clared as a ring in both the Specifications 9 and 10. In the former, it is declared as
a variable of type (ring?) that is defined in the PVS subtheory algebra@ring def

as the Boolean below, whereas in the latter, it was defined as a variable of type ring
that despite being equivalent to the type (ring?) is specified using this Boolean as
ring: NONEMPTY TYPE = (ring?) CONTAINING fullset[T] in the PVS subtheory
algebra@ring.

ring?(S): bool = abelian_group?[T,+,zero](S) AND

semigroup?[T,*](S) AND

left_distributive?[(S)](*,+) AND

right_distributive?[(S)](*,+)

In general, this discrimination is relevant when it is necessary to check that an
arbitrary set satisfies the properties of a specific algebraic structure as in the case
of ring above. The theory ring (as happens similarly in theories field, group,
etc.) contains the assumption that T is a ring, specified as fullset is ring:

ASSUMPTION ring?(fullset[T]) as given in Specification 11, where also the Lemma

plus commutative is specified. This assumption is not mandatory, however it sim-
plifies the specification of some results. For instance, omitting such assumption,
one should quantify variables R of type (ring?), x, y of type (R): FORALL(R:

(ring?), x,y:(R)): x + y = y + x.

Specification 11 Lemma of subtheory ring about commutative property

fullset_is_ring: ASSUMPTION ring?( fullset[T])
x,y: VAR T
plus_commutative: LEMMA x + y = y + x

Consider a hypothetical PVS subtheory hyp theory and suppose one wants to
check in it that for an arbitrary set whose elements have type T, say S : set[T],
ring?(S) holds. For this, one must verify that addition is commutative on S, an
obligation expressed in abelian group?[T,+,zero](S). If subtheory ring were
imported with parameters [T, +, *, zero] by hyp theory, one may be wrongly
assuming that T is a ring, and can apply directly the Lemma plus commutative

(Specification 11). In order to contour this issue, the subtheory ring def can be
imported by hyp theory with the same parameters allowing in this manner the
correct inference of the commutativity of + in S. For this reason, the theory algebra

includes definitional sub-theories (with names using the suffix def) for several
structures such as ring ideal def, ring with one def, and ring nz closed def

(in Figure 2).

3.2.3 Subtheory ring cosets lemmas

In the subtheory cosets def, a coset was specified for setsG andH whose elements
have arbitrary types (see Specification 7). However, in a context where elements
of type T under operations + and ∗ constitute a ring, such as required in the
subtheory ring cosets lemmas, one can formalize a lemma asserting that a set A
is a left coset if and only if it is a right coset. From this, it is possible to establish
a characterization of cosets from left cosets, as shown in Specification 12.

Specification 12 Characterization of cosets from left cosets

fullset_is_ring: ASSUMPTION ring?( fullset[T])
A,S,I : VAR set[T]

lcoset_iff_rcoset: LEMMA
(left_coset?(S,I)(A) IFF right_coset?(S,I)(A))

lcoset_iff_coset: LEMMA
(left_coset?(S,I)(A) IFF coset?(S,I)(A))

The subtheory ring cosets lemmas brings the formalization of other interest-
ing results, such as:

(i) when I is an ideal of a ring R, any member of a coset A of I in R is a generator
of A, that is, A = a+ I, for any a ∈ A;

(ii) if H is a subring and I is an ideal of R then H + I is a subring of R and I is
an ideal in H + I.

The result in Item (i) above is important to establish a characterization of
sum and product of cosets formalized in the subtheory quotient rings, as will

be discussed. The result in item (ii) is widely used in branches of the proof of the
Second Isomorphism Theorem and its auxiliary lemmas.

3.2.4 Kernel of a homomorphism

In the subtheory ring homomorphisms def, the kernel of a homomorphism ϕ be-
tween magmas R1 and R2 was defined as the set of elements of R1 mapped into
some element zeroR2

of R2 by ϕ, as shown in Specification 13.

Specification 13 Kernel of a homomorphism

R_kernel(R1,R2)(phi: R_homomorphism(R1,R2)): set[T1] =
{a:T1 | R1(a) AND R2(zero2) AND phi(a) = zero2}

In the subtheory ring homomorphisms def, since R1 and R2 are not necessarily
rings, one must require that the neutral element exists for +R2

, which is denoted
as zero2 in Specification 13. But, in the subtheory ring homomorphism lemmas, R1

and R2 were considered as rings, structures of the form (R1,+R1
, ∗R1

, zeroR1
) and

(R2,+R2
, ∗R2

, zeroR2
) and, as consequence of this, lemmas involving kernel were

formally proved, among them:

– the kernel of a ring homomorphism ϕ : R1 → R2 is an ideal (and particularly
a ring) in R1 and,

– ϕ is a monomorphism if and only if the kernel of ϕ is the set {zeroR1
}.

The assertion that the kernel is an ideal is used to prove that the isomorphism
presented in the First Isomorphism Theorem is, in fact, a ring homomorphism,
since the quotient R/I is a ring if R is a ring and I is an ideal in R.

3.2.5 Quotient rings

Observing the isomorphism theorems for rings and the characterization of prime
and maximal ideals as described in Theorems 1, 2, 3, 6 and 8, one can no-
tice that quotient rings are relevant algebraic structures involved in such results.
The algebra of quotient rings was specified in the sub-theories cosets def and
product coset def (Figures 1 and 2), as show in Specification 14, where in par-
ticular /(R,I) is a prefix notation for R/I.

Specification 14 The algebraic structure quotient ring

R, I: VAR set[T]

lproduct(R,I)(A,B:left_coset(R,I)) : set[T]
= (lc_gen(R,I)(A) * lc_gen(R,I)(B)) + I

rproduct(R,I)(A,B:right_coset(R,I)) : set[T]
= I + (rc_gen(R,I)(A) * rc_gen(R,I)(B))

product(R,I)(A,B:coset(R,I)) : set[T] = lproduct(R,I)(A,B)

add(R,I)(A,B:coset(R,I)) : set[T]
= (lc_gen(R,I)(A) + lc_gen(R,I)(B)) + I

/(R,I) : setof[set[T]] = {s:set[T] | coset?(R,I)(s)}

In a context where R is a ring, such as occurs in the subtheory quotient rings,
lproduct(R,I)(A) = rproduct(R,I)(A) holds since left and right cosets are the
same (see Specification 14), where A is a coset of an ideal I in R. Other proper-
ties about quotient rings were formalized in the sub-theories quotient rings and
quotient rings with one, among them the ones presented below.

(i) A characterization of sum and product of cosets, denoted as add and product,
respectively: if I is an ideal in a ring R and A = a+ I and B = b+ I are left
cosets of I then add(A,B) = (a+ b) + I and product(A,B) = (a ∗ b) + I.

(ii) A proof that (R/I, add, product, I) is a structure that is a ring, if I is an ideal in
a ring R, and, in particular, that if R is commutative then R/I is a commutative
ring and, if R has an identity oneR then oneR + I is an identity of R/I. These
assertions are crucial since the isomorphism theorems for rings ensure that
there is a ring homomorphism between structures involving quotients.

(iii) A useful lemma to verify proof obligations generated by the Third Isomorphism
Theorem; namely, if I and J are ideals of a ring R then I/J is an ideal of R/J .

(iv) A lemma establishing that if R is a ring with identity oneR and I is an ideal
in R such that I 6= R then I 6= oneR + I. This result is widely used in the
formalization of Theorem 6 in order to guarantee that the quotient R/P has
the identity oneR +P 6= P satisfying a required condition to ensure that R/P
is an integral domain.

4 Formalization of the three isomorphism theorems for rings

This section discusses the PVS sub-theories of rings related with the three iso-
morphism theorems for rings (See Figure 1).

The subtheory ring 1st isomorphism theorem contains the formalization of
the First Isomorphism Theorem as well as a series of auxiliary lemmas presented
as items in Lemma 1.

Lemma 1 (Aux. lemma for the First Isomorphism Theorem) If φ : R→ S
is a homomorphism of rings and I is an ideal of R, which is contained in the kernel
of φ, then there is a homomorphism of rings f : R/I → S such that f(a+I) = φ(a)
for all a ∈ R and:

(i) the image of f is equal to the image of φ;
(ii) ker(f) = ker(φ)/I;

(iii) f is an epimorphism iff φ is an epimorphism;
(iv) f is a monomorphism iff ker(φ) = I;
(v) f is an isomorphism iff φ is an epimorphism and ker(φ) = I.

The formalization of Lemma 1 builds the homomorphism f as f(a+ I) = φ(a)
explicitly and verifies that f is a well-defined function that satisfies the definition
of R homomorphism. Items (i) and (ii) were obtained by proving equality of sets,
whereas item (iii) was verified by the expansion of the definition of R epimorphism.
Item (ii) and the characterization of monomorphims from the kernel were used for
proving item (iv). Lastly, one can obtain item (v) as a natural consequence of
items (iii) and (iv).

The First Isomorphism Theorem was given as shown in Specification 15 and,
roughly speaking, its formalization was obtained by considering the ideal I =

ker(ϕ) in item (v) of Lemma 1. The first four parameters of R isomorphic? in
Specification 15 correspond to the type, the addition and multiplication and the
neutral element for the addition of the quotient ring R/ker(ϕ), whereas the last
four are the parameters for the image of ϕ that is a subring of S.

Specification 15 First Isomorphism Theorem for rings (Theorem 1)

first_isomorphism_th: THEOREM
FORALL(phi: R_homomorphism(R,S)):
R_isomorphic ?[ coset(R, R_kernel(R,S)(phi)),

add(R, R_kernel(R,S)(phi)),
product(R, R_kernel(R,S)(phi)),
R_kernel(R,S)(phi),
D, s, p, zerod]

(R/R_kernel(R,S)(phi), image(phi)(R))

The subtheory ring 2nd 3rd isomorphism theorems brings the formalization
of the Second and Third Isomorphism Theorems. An auxiliary lemma for each one
of these Theorems was required, as presented below.

Lemma 2 (Aux. lemma for the Second Isomorphism Theorem) If H is
a subring and I is an ideal of a ring R then there is an epimorphism ϕ : H →
(H + I)/I such that ker(ϕ) = H ∩ I.

Lemma 3 (Aux. lemma for the Third Isomorphism Theorem) If I and J
are ideals in a ring R and J ⊂ I then there is an epimorphism ϕ : R/J → R/I
such that ker(ϕ) = I/J .

The Second and Third Isomorphism Theorems, given in Specification 16, were
formalized applying the First Isomorphism Theorem with Lemmas 2 and 3, re-
spectively.

Specification 16 Second and Third Isomorphism Theorems for rings (Theorems 2 and 3)

second_isomorphism_th: THEOREM
subring ?(H,R) AND ideal ?(I,R) IMPLIES
R_isomorphic ?[ coset(H,intersection(H,I)),

add(H,intersection(H,I)),
product(H,intersection(H,I)),
intersection(H,I),coset(sum(H,I),I),
add(sum(H,I),I),product(sum(H,I),I),I]

(H/intersection(H,I), sum(H,I)/I)

third_isomorphism_th: THEOREM
(ideal?(I,R) AND ideal?(J,R) AND subset ?(J,I)) IMPLIES
R_isomorphic ?[ coset[coset(R,J),add(R,J)](R/J,I/J),

add[coset(R,J),add(R,J),
product(R,J)](R/J,I/J),
product[coset(R,J),add(R,J),
product(R,J)](R/J,I/J), I/J,
coset(R,I),add(R,I),product(R,I),I]

(/[ coset(R,J),add(R,J),product(R,J)](R/J,I/J),R/I)

5 Formalization of properties about ideals

In this section, the PVS sub-theories of rings presented in Figure 2 and related
with formalizations about principal, prime and maximal ideals are discussed.

5.1 Ideals generated by one element

The subtheory ring one generator brings auxiliary lemmas to formalize, as de-
scribed in the next three subsections, the alternative characterizations of principal
and prime ideals as stated in Theorems 4 and 5 as well as the results about max-
imal ideals as described in Theorems 7 and 8.

Specification 17 presents the main definitions of this subtheory, built from a
specific element a of a ring R. The subset of R defined as the set of elements
generated as {r ∗ a+ a ∗ s+n · a+

∑m
i=1 ri ∗ a ∗ si | r, s, ri, si ∈ R;m ∈ N \ {0};n ∈

Z} is specified as one gen(R)(a). It depends on two functions: (i) the recursive
function R sigma(low,high,F) that adds a number of high− low elements,which
are determined by F : nat→ T; and, (ii) F one gen(R,a,F,G), which performs the
product on the left and right sides of a by members of R chosen by the functions
F and G.

Specification 17 The set one gen(R)(a) generated by the element a of a ring R

F_one_gen(R : ring , a : (R), F,G: [nat -> (R)]):
[nat -> (R)] = LAMBDA (i : nat) : F(i)*a*G(i)

one_gen(R)(a:(R)): set[(R)] =
{t: (R) | EXISTS (r,s: (R), n:int , F,G: [nat ->(R)], m:nat):
t = r*a+a*s+times(a,n) + R_sigma(0,m, F_one_gen(R,a,F,G))}

Among other properties, one gen(R)(a) was proved to be an ideal of the ring
R. Furthermore, if R is commutative or has an identity simpler characterizations
of one gen(R)(a) were developed as those given in Specifications 18 and 19.

Specification 18 Properties about one gen and a simpler characterization for one gen(R)(a),
when R is commutative

one_gen_is_ideal:LEMMA
FORALL(R: ring ,a:(R)): ideal?(one_gen(R)(a),R)

commutative_one_gen(R)(a:(R)): set[(R)] =
{t: (R) | EXISTS (r:(R), n:int): t = r*a+times(a,n)}

commutative_one_gen_charac: LEMMA
FORALL(R:( commutative_ring?),a:(R)):
one_gen(R)(a) = commutative_one_gen(R)(a)

Specification 19 A simpler characterization for one gen(R)(a), when R is commutative and
has an identity, as part of the subtheory ring with id one generator

R : VAR (ring_with_one?)

commutative_id_one_gen_charac: LEMMA
FORALL(a:(R)): commutative_ring?(R) IMPLIES
one_gen(R)(a) = +[T,*](a,R)

5.2 Principal ideals

There are two sub-theories concerning principal ideals:

– ring principal ideal def that contains the definition of principal ideals from
the family of ideals in a ring R containing a specific element a ∈ R, specified
as fam ideal(R:(ring?))(a:(R)), and considering the intersection of all such
ideals (see Specification 20);

– ring principal ideal that brings several properties, having as main result
the formalization of the characterization of prime ideals from the constructor
one gen(R)(a) that establishes a shape for the elements of principal ideals as
enunciated in Theorem 4 (i) (see Specification 21).

Specification 20 Definition of principal ideals

R : VAR set[T]

fam_ideal(R:(ring ?))(a:(R)): setof[set[(T)]] =
{A: ideal(R) | member(a,A)}

principal_ideal(R:(ring ?))(a:(R)): set[T] =
Intersection(fam_ideal(R)(a))

Specification 21 Characterization of principal ideals from one gen(R)(a)

principal_ideal_charac: LEMMA
FORALL(R: (ring?), a:(R)):
principal_ideal(R)(a) = one_gen(R)(a)

The proofs of the items (ii) and (iii) of Theorem 4 are obtained combining the
Lemma principal ideal charac with Lemmas commutative one gen charac and
commutative id one gen charac, respectively (Specifications 18 and 19).

5.3 Prime ideals

The subtheory ring prime ideal def gives the specification of prime ideals in a
ring R defining the type prime ideal (see Specification 22).

Specification 22 Definition of prime ideals

IMPORTING cosets_def[T,*]

R: VAR (ring?)
A, B, P: VAR set[T]

prime_ideal ?(P,R): bool =
ideal?(P,R) AND P/=R AND

FORALL (A,B:ideal(R)): subset ?(sum(A,B),P)
IMPLIES subset ?(A,P) OR subset ?(B,P)

prime_ideal(R): TYPE = {P: set[T] | prime_ideal?(P,R)}

The formalization of Theorem 5 is in the subtheory ring prime ideal. This
theorem is specified as lemmas prime ideal prop1 and prime ideal prop2 (see
Specification 23). These lemmas establish a useful characterization of prime ideals
in commutative rings. The formalization of both these lemmas is mainly based on
case analysis on the definitions, as given in algebra textbook proofs. In addition

for proving the second lemma it is necessary to apply a corollary (inferred from
lemmas in Specification 18) that states that commutative one gen(R)(a), for any
a in R, is an ideal in R.

Specification 23 Characterization of prime ideals in commutative rings

R: VAR ring
P: VAR set[T]
a,b: VAR T

prime_ideal_prop1: LEMMA
(ideal?(P,R) AND P/=R AND
FORALL (a,b: (R)): member(a*b,P)
IMPLIES member(a,P) OR member(b,P))

IMPLIES prime_ideal?(P,R)

prime_ideal_prop2: LEMMA
commutative_ring?(R) AND prime_ideal?(P,R) IMPLIES
FORALL (a,b: (R)): member(a*b,P)
IMPLIES member(a,P) OR member(b,P)

The subtheory ring with one prime ideal contains the proof of Theorem 6
formalized as Lemma prime ideal charac presented in Specification 24. In this
specification monad?, taken from the theory algebra, is a unary predicate for check-
ing whether the set given as argument is closed under operation ∗, and includes the
element one, which behaves as multiplicative identity over the whole input set. The
formalization of this Lemma depends mainly on Lemma prime ideal nz closed

(also in Specification 24), which states that for a commutative ring R and a prime
ideal P in R, the quotient ring R/P is closed under the product, when the neutral
element for the addition is removed. This means that the set R/P \ {P} is closed
under the product operation defined in subtheory product coset def (see Speci-
fication 14). Furthermore, for specifying Theorem 6 it was necessary to define the
unary predicate integral domain w one that is nothing more than the usual no-
tion of integral domain found in abstract algebra textbooks, and in particular the
one found in [20]. This predicate holds for commutative rings with oneR 6= zeroR
and without zero divisors. In the theory algebra this notion does not impose the
constraint oneR 6= zeroR.

Specification 24 Characterization of prime ideals in commutative rings with identity

R: VAR ring_with_one[T,+,*,zero ,one]
P: VAR set[T]

prime_ideal_nz_closed: LEMMA
(commutative_ring?(R) AND prime_ideal?(P,R)) IMPLIES
nz_closed?
[coset[T,+](R,P),add[T,+,*](R,P),product[T,+ ,*](R,P),P]
(R/P)

prime_ideal_charac: LEMMA
(commutative_ring?(R) AND monad?[T,*,one]( remove(zero ,R))
AND ideal?(P,R)) IMPLIES

(prime_ideal?(P,R) IFF
integral_domain_w_one?[coset[T,+](R,P),add[T,+ ,*](R,P),

product[T,+ ,*](R,P),
P,+[T,+](one ,P)]

(R/P))

5.4 Maximal ideals

In the subtheory ring maximal ideal def, the definition of and the type of max-
imal ideal were specified, as shown in Specification 25.

Specification 25 Definition of maximal ideals

R: VAR (ring?)
M,N: VAR set[T]

maximal_ideal ?(M,R): bool =
ideal?(M,R) AND M/=R AND
FORALL(N:ideal(R)): subset ?(M,N) AND subset ?(N,R) IMPLIES
N = M OR N = R

maximal_ideal(R): TYPE = {M: set[T] | maximal_ideal?(M,R)}

In the subtheory ring maximal ideal, Theorem 7 was specified (See Specifi-
cation 26) and, since for every ring R with identity it holds that R ∗R = R, it was
proved that maximal ideals in commutative rings with identity are prime ideals as
illustrated in the Specification 27.

Specification 26 Maximal versus prime ideals on commutative rings

maximal_prime_ideal: LEMMA
FORALL (R:( commutative_ring?), M: maximal_ideal(R)):
sum[T,*](R,R) = R IMPLIES prime_ideal?(M,R)

Specification 27 Maximal versus prime ideals on commutative rings with identity

R: VAR ring_with_one
M: VAR set[T]

ring_one_maximal_prime_ideal: LEMMA
commutative_ring?(R) AND maximal_ideal?(M,R) IMPLIES
prime_ideal ?(M,R)

Theorem 8 was fully formalized as Lemmas maximal ideal quot field and
quot div ring maximal ideal; in this manner, the characterization of maximal
ideals by quotient rings was established as shown in Specifications 28 and 29.
The characterization of principal ideals given in Theorem 4 is essential to formal-
ize Lemmas maximal prime ideal and maximal ideal quot field. The latter is
proved by firstly (i) ensuring that R/M is a commutative ring with identity with-
out zero divisors by Lemmas maximal prime ideal and prime ideal nz closed

and secondly by (ii) establishing that every element a+M 6= M , a ∈ R, of R/M
has a multiplicative inverse. Technical details can be seen in [20] and will be omit-
ted, however one can point out that it is crucial to consider the ideal M + (a)
generated from the principal ideal (a) and the shape of its elements to show this
property.

Specification 28 Characterization of maximal ideals by quotient rings

maximal_ideal_charac: LEMMA
(commutative_ring?(R) AND ideal?(M,R)) IMPLIES
(maximal_ideal ?(M,R) IFF

division_ring?[coset[T,+](R,M),add(R,M),product(R,M),
M,+[T,+](one ,M)](R/M))

Specification 29 Auxiliary lemmas to characterize maximal ideals by quotient rings

maximal_ideal_quot_field: LEMMA
FORALL(M: maximal_ideal(R)):
commutative_ring?(R) IMPLIES
field?[ coset[T,+](R,M),add(R,M),product(R,M),

M,+[T,+](one ,M)](R/M)

quot_div_ring_maximal_ideal: LEMMA
FORALL(M:ideal(R)):
division_ring?[coset[T,+](R,M),add(R,M),product(R,M),

M,+[T,+](one ,M)](R/M) IMPLIES
maximal_ideal ?(M,R)

6 Algebraic version of CRT

This section discusses the formalization of the Chinese Remainder Theorem for
rings, formalized as consequence of the First Isomorphism Theorem. The sub-
theories related with this part of the formalization of the theory rings are given
in Figure 3, where also the subtheory including the formalization of CRT for
integers, to be discussed in the next section, appears.

quotient rings

rings chinese remainder theorem Z

chinese remainder theorem rings

prop primes extra
ring 1st isomorphism theorem product finseq sets ring

cartesian product quot ring

quotient rings with one

comaximal finseqs ideals

sigma R below

cartesian product finite

ring homomorphisms def

algebra

lcm

lagrange scaf

integral domain with one def

ring with one ideal

ring ideal

ring with one

field def

ring with one def

ring

ring zn

ring cosets lemmas

integral domain def

Fig. 3 Hierarchy of the sub-theories for the Chinese Remainder Theorems for the theory of
rings and the ring of integers

In the literature, a general algebraic version of CRT for arbitrary rings and
ideals is classically given as enunciated in Theorem 9 (e.g., see [12]).

Theorem 9 (Chinese Remainder Theorem: general algebraic version)
Let R be a ring with identity and A1, A2, . . . , Ak ideals in R. The map

ϕ : R → R/A1 × . . .×R/Ak

r 7→ (r +A1, . . . , r +Ak)

is a ring homomorphism with kernel A1 ∩ . . . ∩ Ak. If the condition Ai + Aj = R
holds for each i, j ∈ {1, . . . , k} with i 6= j, which is called comaximality, then ϕ is
surjective and the rings R/(A1∩ . . .∩Ak) and R/A1× . . .×R/Ak are isomorphic.

The general algebraic version of the CRT for rings as enunciated in Theorem 9
was fully formalized and the three crucial steps to obtain such proof are described
in the remaining of this section.

Step 1. Prove that ϕ, as given in the statement of the theorem, is a ring homo-
morphism with kernel A1 ∩ . . . ∩Ak, as given in Specifications 30, 31 and 32.

The notion of Cartesian product is given in Specification 30, part of the subthe-
ory Cartesian product finite. The constructor Cartesian product n receives as
input fsA, a finite sequence of sets of type T, and gives as output the set of finite
sequences of length the same as the length of fsA, of elements of type T. An element
fsz of such set of sequences of length fsA is such that its ith components belongs
to the ith set in the sequence fsA. In PVS, finite sequences of type T are data
structures that consist of their length, say n, and a function from below[n] to
T; for instance, (# length := n, seq := (LAMBDA (i:below[n]): i) #) is the
sequence of the first n naturals.

Specification 30 Constructor for formalization of finite Cartesian product

fsz: VAR finseq[T]
fsA: VAR finseq[set[T]]

Cartesian_product_n(fsA): set[finseq[T]] =
IF length(fsA) = 0 THEN emptyset
ELSE {fsz | length(fsz) = length(fsA) AND

FORALL (i: below(length(fsz ))): member(fsz(i), fsA(i))}
ENDIF

The technical elements required to formalize that R/A1 × . . . × R/Ak is in-
deed a ring are contained in the sub-theories Cartesian product finite and
Cartesian product quot ring, see Specification 31. In a context where elements of
type T under operations + and * and constant zero constitute a ring, the predicate
fsRI(R) checks if a finite sequence of sets of type T are ideals in R; and the con-
structor fsQ(R) builds the finite sequence of quotient rings R/fsA(i) for the input
fsA, where fsRI?(R)(fsA) holds. Thus, Cartesian product n(fsQ(R)(fsA)) is
the Cartesian product of such quotients, and the constructors Sfs and Pfs specify
the operations of addition and multiplication for this Cartesian product. Finally,
that this Cartesian product is a ring is the main result formalized as Lemma
Cartesian product quot ring is ring in Specification 31. The formalization of
this fact required proving that such addition and multiplication operators satisfy
the properties of rings for Cartesian product n(fsQ(R)(fsA)), such as closure of
the operators, existence of the identity for the addition (that is the sequence of
ideals), property of additive inverses, distributivity, etc.

Specification 31 Addition and multiplication (Sfs(fsA) and Pfs(fsA), resp.) for the ring of
Cartesian product of quotient rings

R: VAR ring[T,+,*,zero]
fsA: VAR finseq[set[T]]

% Predicate for sequence of ideals

fsRI?(R)(fsA): bool = FORALL (i: below[length(fsA )]): ideal?(fsA(i), R)
% Type predicate for sequences of ideals

fsI(R): TYPE = {fsA: finseq[set[T]] | fsRI?(R)(fsA)}
% Sequence of quotient rings

fsQ(R)(fsA: fsI(R)): finseq[setof[set[T]]] =
IF length(fsA) = 0 THEN empty_seq
ELSE (# length := length(fsA),

seq := (LAMBDA (i:below[length(fsA )]): R/fsA(i)) #)
ENDIF

% Addition of elements of the Cartesian product of quotient rings
Sfs(R)(fsA:fsI(R))(fsx ,

fsy:( Cartesian_product_n(fsQ(R)(fsA )))):
finseq[set[T]] =

IF length(fsA) = 0 THEN empty_seq
ELSE (# length := length(fsA),

seq := (LAMBDA (i: below[length(fsA )]):
add(R,fsA(i))(fsx(i), fsy(i))) #)

ENDIF
% Product of elements of the Cartesian product of quotient rings

Pfs(R)(fsA:fsI(R))(fsx ,
fsy:( Cartesian_product_n(fsQ(R)(fsA )))):

finseq[set[T]] =
IF length(fsA) = 0 THEN empty_seq
ELSE (# length := length(fsA),

seq := (LAMBDA (i: below[length(fsA )]):
product(R,fsA(i))( fsx(i), fsy(i))) #)

ENDIF

Cartesian_product_quot_ring_is_ring: LEMMA
FORALL (fsA: fsI(R)): length(fsA) /= 0 IMPLIES

ring?( Cartesian_product_n(fsQ(R)(fsA )))

The map ϕ, given in Specification 32, was proved to be a homomorphism
by checking that R homomorphism?(R,cartesian product n(fsQ(R)(fsA)))(phi)

holds, which is done by verification of the conditions of ring homomorphism. The
characterization of the kernel of ϕ as the intersection A1 ∩ . . . ∩ Ak (last line in
Specification 32) was established by extensional equality on the sets (i.e., for all x,
x ∈ kernel(ϕ) if and only if x ∈ A1∩ . . .∩Ak), observing that the neutral element
for addition over the Cartesian product of the quotient R/A1 × . . .×R/Ak is the
sequence of ideals (A1, . . . , Ak). This result requires application of a key lemma
for subrings (lemma self coset in theory ring cosets lemmas, see Fig. 1) that
states that if A is a subring of R then for all x ∈ R, x+A = A iff x ∈ A. Thus, the
formalization of the equality proceeds in both directions according to the following
argumentation: by definition of ϕ, ϕ(x) = (x+A1, . . . , x+Ak), and by definition
of kernel, x ∈ kernel(ϕ) iff ϕ(x) = (A1, . . . , Ak). Consequently, x ∈ Ai for all
i ∈ {1, . . . , k} and so x ∈ A1 ∩ . . . ∩Ak.

Specification 32 ϕ is a homomorphism with kernel A1 ∩ . . . ∩Ak

CRT_aux _1: LEMMA
FORALL (fsA: fsI(R) | length(fsA) /= 0):
LET phi = LAMBDA (x: (R)): (# length := length(fsA),

seq := (LAMBDA (i: below[length(fsA )]): x + fsA(i)) #) IN
R_homomorphism ?[T,+,*,zero ,

(cartesian_product_n[set[T]](fsQ(R)(fsA))), Sfs(R)(fsA),
Pfs(R)(fsA),fsA]

(R,cartesian_product_n[set[T]](fsQ(R)(fsA )))( phi)
AND

R_kernel[T,+,*,zero ,
(cartesian_product_n[set[T]](fsQ(R)(fsA))), Sfs(R)(fsA),
Pfs(R)(fsA),fsA]

(R,cartesian_product_n[set[T]](fsQ(R)(fsA )))( phi)= Intersection(seq2set(fsA))

Step 2. Prove that ϕ is surjective. This step uses the hypothesis that the ideals
A1, . . . , Ak are pairwise comaximal (i.e., Ai + Aj = R when i 6= j). This proof
is constructive and the interesting substeps and technical details are explained
below.

The concepts of comaximal and sequence of pairwise comaximal ideals were
defined as in Specification 33.

Specification 33 Definition of comaximal ideals

comaximal_ideals ?(R)(M,N): bool =
ideal?(M,R) AND ideal ?(N,R) AND sum(M,N) = R

fsICM?(R)(fsA): bool = fsRI?(R)(fsA) AND
(FORALL (i,j: below[length(fsA )]):

i /= j IMPLIES comaximal_ideals ?(R)(fsA(i),fsA(j)))

fsICM(R): TYPE = {fsA: finseq[set[T]] | fsICM?(R)(fsA)}

The key to prove the surjectivity of ϕ, Lemma surjective aux 1 in Specifica-
tion 34, is to build a sequence s = (s(1), . . . , s(k)) over R such that s(i) ∈ Aj for all
j 6= i and s(i)+Ai = one+Ai . The elements of this sequence are characterized as
s(i) =

∏i−1
j=1(one−xj)∗

∏k
j=i+1(one−xj), where xj ∈ Ai and yj = (one−xj) ∈ Aj

for each j = 1, . . . , i− 1, i+ 1, k. Notice that, the sequence s is well-defined since
the ideals Ai and Aj are comaximal for i 6= j and the ring R has identity; conse-
quently, there exist elements xi ∈ Ai and yj ∈ Aj such that one = xi+yj whenever
i 6= j. By performing calculations over s(i), since Ai is closed under addition and
multiplication, one can infer that s(i) = one + a, where a ∈ Ai. Consequently,
s(i) +Ai = one+ (a+Ai) = one+Ai. Now, it is not difficult to establish that the
map ϕ is surjective. Consider y = (r1+A1, . . . , rk+Ak) ∈ R/A1×. . .×R/Ak, where
ri ∈ R, 1 ≤ i ≤ k, the sequence s as defined before and x =

∑k
i=1 s(i) ∗ ri ∈ R.

Fixing an index, one can infer that
(∑k

i=1 s(i) ∗ ri
)

+Ai = (s(i) ∗ ri) +Ai, since

s(j) ∈ Ai for i 6= j. Consequently,
(∑k

i=1 s(i) ∗ ri
)

+Ai = (one+Ai)∗ (ri +Ai) =

ri +Ai and ϕ(x) = y.

Specification 34 Auxiliary lemma to prove the surjectivity of ϕ

surjective_aux_1: LEMMA
FORALL (R: ring_with_one , fsA: fsICM(R)):

EXISTS (s: finseq [(R)] | length(s) = length(fsA )):
FORALL (i: below[length(fsA )]):

s(i) + fsA(i) = one + fsA(i) AND
(FORALL (j: below[length(fsA)] | j /= i): member(s(i),fsA(j)))

The sequence s needed to formalize the Lemma surjective aux 1 was spec-
ified from the set oneSet(R,fsA) defined from a ring R and a sequence of sets
fsA such that for all 0 ≤ i, j ≤ length(fsA) − 1, oneSet(R,fsA)(i,j) = {x ∈
fsA(i)|(one − x) ∈ fsA(j), whenever j 6= i} and a recursive function that mul-
tiplies elements of a given sequence of elements of arbitrary type; namely, the
function product fs rec(fs) in Specification 35. Specifically, to construct s, the
sequence fs in product fs rec(fs) has length k−1 and was built from the choose
operator in the native library of PVS, which specifies the axiom of choice. Thus,
for each 1 ≤ j ≤ k, j 6= i, fs(i) = one−choose(oneSet(R, fsA)(i, j)). Obviously, a

proof obligation regarding the non emptiness of oneSet(R,fsA)(i,j) arises which
was formalized in the Lemma oneSet nonempty by considering the sequence of sets
fsA as comaximal ideals (See Specification 36).

Specification 35 Function that multiplies the elements of a sequence

fs: VAR finseq[T]

product_fs_rec(fs)(i: below[length(fs)]): RECURSIVE T =
IF i = 0 THEN seq(fs)(0)
ELSE seq(fs)(i) * product_fs_rec(fs)(i-1)
ENDIF MEASURE i

Specification 36 Non emptiness of the set oneSet(R,fsA)(i,j) for a sequence of comaximal
ideals

fsA: VAR finseq[set[T]]

oneSet_nonempty: LEMMA
FORALL (R: ring_with_one , fsA: fsICM(R) | length(fsA) > 1,

i: below[length(fsA)], (j: below[length(fsA)] | j /= i)):
NOT empty?( oneSet(R, fsA)(i,j))

Step 3. Prove that R/(A1 ∩ . . . ∩Ak) and R/A1 × . . .×R/Ak are isomorphic.

Specification 37 Chinese Remainder Theorem for rings

Chinese_Remainder_Theorem: LEMMA
FORALL (R: ring_with_one , fsA: fsICM(R) | length(fsA) /= 0):

R_isomorphic ?(R/Intersection(seq2set(fsA)),
Cartesian_product_n(fsQ(R)(fsA )))

Now, one can notice that Chinese Remainder Theorem (see Specification 37)
is a direct consequence of the First Isomorphism Theorem. In fact, in Theorem 1
one must consider the ring homomorphism

ϕ : R → R/A1 × . . .×R/Ak

r 7→ (r +A1, . . . , r +Ak)

Thus, by the Step 1 one can ensure that ker(ϕ) = A1 ∩ . . . ∩ Ak and by the Step
2 that the image of ϕ is R/A1 × . . .×R/Ak.

7 Formalization of the number-theoretical CRT

This section shows how the general version of CRT for the theory of rings is
applied to formalize the CRT for the structure of integers (see Figure 3, subthe-
ory chinese remainder theorem Z). This is a well-known crucial application that
shows how formalized abstract results in the theory of rings are applied to obtain
important properties about numerical structures. Besides, it is illustrated how
CRT for integers can be applied to obtain numerical properties by explaining how
the well-known result that “the product of two positive integers equals the product
of their greatest common divisor and least common multiple” is formalized.

CRT for integers is well-known as a classical result in number theory with
applications in computing and coding [11], security [17] [11], signal processing [25]
and other fields. The number-theoretical version of CRT states that an integer
number can be characterized by a sequence of its remainders modulo some integers
and, basically, this consists in solving a system of congruences. From the point of
view of algebraic structures, the standard version of CRT can be formulated in a
more general setting as a ring isomorphism, as shown in Theorem 10.

Theorem 10 (CRT: number-theoretical version for Z) Let m1,m2, . . . ,mk

be positive integers such that mi and mj are coprime for i 6= j and let m =
m1 ∗m2 ∗ . . . ∗mk. Then there is an isomorphism between the rings Zm and the
Cartesian product Zm1 × . . .× Zmk .

Specification 38 contains the Corollary chinese remainder th for int that is
the CRT for Z. The Corollary is preceded by the main results used in its proof;
indeed, lemmas nz fs intersection, nZ ideal and, nZ mZ comaximal.

The number-theoretical version of the CRT, Theorem 10, is obtained as a
corollary of the general algebraic version of CRT, Theorem 9. Before instantiating
the algebraic version of CRT, lemma nZ fs intersection is applied. In sequence,
Theorem 9 is instantiated with the ring of integers, Z, and the ideals miZ in Z,
where 1 ≤ i ≤ k and mi ∈ Z and, for guaranteeing typing properties of this
instantiation of the algebraic version of CRT, the other two lemmas, nZ ideal

and nZ mZ comaximal, are applied. Since the quotient Z/miZ is Zmi (the ring of
integers modulo mi), one can concluded that there is an isomorphism between the
rings Zm and the Cartesian product Zm1 × . . .×Zmk , where m = m1 ∗m2 . . .mk.

Instantiating Theorem 9 as before, one can initially infer that there is a homo-
morphism from Z to Zm1× . . .×Zmk . Furthermore, for arbitrary integers m and n,
mZ+nZ = dZ, where d is the greatest common divisor of m and n. Consequently,
if m1,m2, . . . ,mk are positive integers such that mi and mj are coprime for i 6= j,
miZ +mjZ = Z for each i, j ∈ {1, . . . , k} with i 6= j. This shows that there is an
isomorphism between the rings Zm and Zm1 × . . . × Zmk , where m is as above,
since m1Z∩ . . .∩mkZ = mZ. Thus, the number-theoretical version of the CRT is
concluded.

Specification 38 Chinese Remainder Theorem for integers

nZ(n): set[int] = {x | EXISTS k: x=n*k}

fs_rel_primes(fsn): bool =
FORALL (i,j: below[length(fsn )]): i /= j IMPLIES rel_prime(fsn(i),fsn(j))

Z_commutative_ring_w_one: LEMMA commutative_ring_with_one?[int ,+,*,0,1](Z)

nZ_ideal: LEMMA FORALL n: ideal?(nZ(n),Z)

Z(n): setofsets[int] = Z/nZ(n)

Zn_card_n: LEMMA card(Z(n)) = n

nZ_mZ_sum: LEMMA sum(nZ(n),nZ(m)) = nZ(gcd(n,m))

nZ_mZ_comaximal: LEMMA comaximal_ideals ?(Z)(nZ(n),nZ(m)) IFF rel_prime(n,m)

nZ_mZ_intersection: LEMMA intersection(nZ(n),nZ(m)) = nZ(lcm(n,m))

nZ_mZ_rel_prime_intersection: COROLLARY
rel_prime(n,m) IMPLIES intersection(nZ(n),nZ(m)) = nZ(n*m)

nZ_fs_intersection: LEMMA
(length(fsn) /= 0 AND fs_rel_primes(fsn))

IMPLIES
LET fsnZ: finseq[set[int]] =

(# length := length(fsn),
seq:= (LAMBDA (i:below[length(fsn )]): nZ(fsn(i))) #) IN

Intersection(seq2set(fsnZ)) = nZ(product(fsn))

chinese_remainder_th_for_int : COROLLARY
FORALL (fspn: finseq[posnat] | length(fspn) /= 0 AND fs_rel_primes(fspn)) :

R_isomorphic ?(Z/nZ(product(fspn)),
cartesian_product_n[set[int]]( fsQ(Z)(fsnZ(fspn ))))

gcd_lcm_property: COROLLARY lcm(n,m) = n*m/gcd(n,m)

Despite the fact that commutativity of Z is used in this formalization for al-
gebraic simplifications, which are not always essential, it is important to stress
here that the crucial use of commutativity of Z appears in the formalization
of Lemma nZ mZ intersection. Lemma nZ mZ intersection is used for proving
Corollary nZ mZ rel prime intersection and Lemma nZ fs intersection (see
Specification 38). Indeed, in order to prove the inclusion lcm(m,n)Z ⊂ nZ ∩mZ,
required to prove nZ mZ intersection, one can proceed as following. Consider
x ∈ lcm(n,m)Z, which means that x = lcm(n,m) ∗ z for some z ∈ Z. Since
there are integers k and l such that l ∗ m = lcm(n,m) = k ∗ n, one has that
x = l ∗ m ∗ z = k ∗ n ∗ z. And since the ring of integers is commutative, one
has x = m ∗ (l ∗ z) and x = n ∗ (k ∗ z). Consequently, x ∈ nZ ∩ mZ. Thus,
commutativity of Z is required to guarantee that when m and n are coprime,
mZ ∩ nZ = (m ∗ n)Z, and this is used for proving the required generalization in
Lemma nZ fs intersection: m1Z ∩ . . . ∩mkZ = (m1 ∗ . . . ∗mk)Z.

This section is closed illustrating the use of CRT for Z (Theorem 10) to for-
malize lcm(n,m)∗gcd(n,m) = n∗m, for n and m positive integers (See Corollary
gcd lcm property in Specification 38). This result is obtained instantiating The-
orem 10 with the sequence of relative primes ( n

gcd(n,m) ,
m

gcd(n,m) ) obtaining the
isomorphism between the structures below.

Z/( n

gcd(n,m)
∗ m

gcd(n,m)
)Z ' Z n

gcd(n,m)
× Z m

gcd(n,m)

Notice that, since the integers n
gcd(n,m) and m

gcd(n,m) are relative primes, ap-

plying lemmas nZ fs instersection, nZ mZ rel prime intersection (in Specifi-
cation 38) one obtains:(

n

gcd(n,m)
∗ m

gcd(n,m)

)
Z =

n

gcd(n,m)
Z
⋂ m

gcd(n,m)
Z

Then, applying lemma nZ mZ intersection (also, in Specification 38), one ob-
tains:

n

gcd(n,m)
Z
⋂ m

gcd(n,m)
Z = lcm

(
n

gcd(n,m)
,

m

gcd(n,m)

)
Z

Consequently, the finite structures below are isomorphic and have the same
cardinality.

Z/
(
lcm(

n

gcd(n,m)
,

m

gcd(n,m)
)

)
Z ' Z n

gcd(n,m)
× Z m

gcd(n,m)

Since Zn is defined as Z/nZ, for all positive integer n and the cardinality of
Zn is n (see Z(n) and lemma Zn card n in Specification 38), using properties of
finite Cartesian product, one obtains that

lcm(
n

gcd(n,m)
,

m

gcd(n,m)
) =

n

gcd(n,m)
∗ m

gcd(n,m)

Finally, since lcm(d∗ i, d∗ j) = d∗ lcm(i, j), for positive integers d, i and j, one
concludes the desired result.

1

gcd(n,m)
∗ lcm(n,m) =

n

gcd(n,m)
∗ m

gcd(n,m)

8 Related work

Emmy Noether introduced the isomorphism theorems for modules in her seminal
work [27] giving rise to an interpretation of such theorems as general laws of
isomorphisms for several algebraic structures such as groups, rings and fields [36].

Regarding previous PVS work in theories algebra ([6]) and rings ([34]), dis-
cussed in the introduction, the present formalization adapts and substantially
extends part of rings using algebra as a basis. All previous results about groups
were maintained, while those in theory rings were adapted trying to obtain defini-
tions as general as possible. An interesting feature of the generality is expressed in
the specification of the notion of homomorphism, now given for the simple struc-
ture of magmas in such a manner that it can be applied to several other more
elaborated algebraic structures, including monoids, semigroups, groups, rings as
well as fields.

8.1 Relation with other formalizations

There are several formalizations containing a wide range of theorems for several
abstract algebra structures. In Coq results about groups, rings and ordered fields
were formalized as part of the FTA project [14]; further related formalizations
in Coq deal with finite group theory [16] culminating in the formalization of the
Feit and Thompson’s proof of the Odd Order Theorem that states that every finite
group of odd order is solvable [15]. Also in Coq, formalizations of real ordered fields
[9] and finite fields (to check eliptic curve cryptography) [29], and a formalization
of rings with explicit divisibility [8] are available. In Nuprl and Mizar it is provided
a formal proof of the Binomial Theorem for rings, [21,32], respectively. In ACL2
a hierarchy of algebraic structures ranging from setoids to vector spaces is built
focusing on the formalization of computer algebra systems [18]. Lean’s library
mathlib, also contains quite a bit of abstract algebra. It contains a definition
of the First Isomorphism Theorem for groups, but no definition of isomorphism
theorem for rings. It includes also a definition of an abstract version of the Chinese

Remainder Theorem for commutative rings, but without formalization of its proof
[10].

The Algebra Library of Isabelle/HOL [1] provides a wide range of theorems
on mathematical structures, including results on rings, groups, factorization over
ideals, rings of integers and polynomial rings. This library contains formalizations
of the First and Second Isomorphism theorems for groups and more recently incor-
porated also a Formalization of the First Isomorphism Theorem for rings as well
as an algebraic version of the Chinese Remainder Theorem for commutative rings
with identity [1]. A formalization of the First Isomorphism Theorem for rings is
also available in Mizar [23] (which, as ACL2, is a first-order set theoretical based
framework). This formalization differs from the one given here in which the flex-
ibility of higher order logic, basis of the PVS deductive engine, gives rise in a
natural manner to specifications and formalizations that are close to the ones in
textbooks on algebra.

Regarding the Chinese Remainder theorem, the first known computerized for-
malization of its number-theoretical version was performed using the inductive en-
gine of the Rewrite Rule Laboratory RRL by Zhang and Hua [38]. After that, other
formalizations of this theorem, also from the number-theoretical perspective, i.e.,
for the specific commutative rings of integers, have been reported. Schwarzweller
discusses different aspects of formalizations of the CRT in Mizar, HOL Light, hol98
and Coq [33]. Also, number-theoretical formalizations of CRT in ACL2 and Veri-
Fun have been briefly discussed, respectively, by Russinoff in [31] and Walther in
[37].

In contrast with all these number-theoretical formalizations, both the Isabelle
proof given by Aransay et al. in [1] and the one reported in this paper are for-
malizations of algebraic versions of this theorem. Ballarin in [4] explains how the
locales of Isabelle were explored in order to provide the concise formalizations of
the algebraic notions and results in [1], which cover the first part of Jacobson’s
textbook on Basic Algebra [22]. Differently from the formalization presented in this
paper, the one in [1] includes the additional hypothesis of commutativity, that is
indeed unnecessary. This hypothesis would be required only if one wants to prove
that the ring R/(A1 · . . . ·Ak) is isomorphic to the ring R/(A1 ∩ . . . ∩Ak) as well
as to the ring R/A1 × . . . × R/Ak, as stated in versions of this theorem as those
given in Dummit and Foote’s [12] and Hungerford’s books [20]. The operator ·
above refers to the set of linear combinations of products of elements of each ideal,
i.e., A1 · . . . · Ak = {

∑n
1 xi1 ∗ . . . ∗ xik | n ∈ N, xij ∈ Aj , for 1 ≤ j ≤ k}. Indeed,

this isomorphism is a consequence of the identity A1 · . . . · Ak = A1 ∩ . . . ∩ Ak,
which depends on the commutativity. Also, notice that this holds of course for the
number theoretical version of the CRT since Z is commutative.

8.2 Formalization versus pen-and-paper proofs

We discuss some interesting aspects of the formalization regarding the pen-and-
paper proofs as they are presented in algebra textbooks such as Hungerford’s, and
Dummit and Foote’s textbooks [20], [12].

First, several concepts were formalized in a more general manner than in text-
books. For instance, in Hungerford’s textbook, the definitions of left and right
cosets are stated for groups. Following this approach, in the subtheory cosets

of the theory algebra from nasalib, the definition was specified over a group
(G,*,one) and a subgroup H of G as:

left coset(G:group,H:subgroup(G))(a:(G)): { s: set[T] | subset?(s,G)} = a*H.
We generalize this specification of left and right cosets over sets of type set[T]

with a binary operation + as:
left coset?(G,H:set[T])(A:set[T]) : bool = (EXISTS (a: (G)): A = a+H).
In this manner it is possible to apply this definition over other structures

different from groups, thus reducing the number of proofs obligations generated
by PVS since it is not necessary to check that G and H are respectively a group
and a subgroup.

Second, some proofs were modified. The specification of the CRT for ring theory
follows the presentation by Dummit and Foote’s in [12] but the formalization differs
of the proof presented in such textbook in some aspects. For example, Dummit and
Foote consider the theorem for a ring R with identity one 6= zero excluding the zero
ring as a possible case, which is not required in the formalization. Also, Dummit
and Foote’s proof considers only the case of two ideals, arguing that the general
case can be verified easily by induction. Formalizing CRT for rings following this
approach, if possible, would have a higher cost than our direct proof. For example,
in order to formalize that the map

phi = LAMBDA (x: (R)): (# length := length(fsA),
seq := (LAMBDA (i: below[length(fsA)]): x + fsA(i)) #)

defined from a ring R to the Cartesian product R/A1 × . . .× R/Ak over a sequence
fsA = (A1, . . . , Ak) of ideals (see Specification 32) is a homomorphism, one must
verify, for instance, that for all j ≤ k, and a, b in R, (a+b)+Aj = (a+Aj)+(b+Aj)
holds. It can be easily verified, for all j ≤ k, applying a lemma that establishes
a characterization for addition of cosets of a quotient ring. Notice that, it has an
equivalent cost of the analysis for two ideals in a proof by induction, where k = 2.
Besides, in the induction step, the analysis given for two ideals cannot be replied
in a straightforward manner since one has to build structures such as an ideal A
such that (R/A1 × . . .×R/An) ' R/A, to be able to apply the reasoning for two
ideals to conclude that the map phi is a homomorphism from R to R/A×R/An+1.

Finally, it is worth highlighting the difference between the formalization in
PVS and the version presented in Hungerford’s textbook of the following result
about homomorphism of rings:

Lemma 4 (Items 2 and 4, Corollary 2.21 in [20]) If R is a ring with oneR 6=
zeroR then R has no proper ideals iff every nonzero homomorphism of rings from
R to S is a monomorphism.

R has no proper ideal means that the only ideals of R are R and {zeroR}.
Hungerford argues that if I 6= R then the canonical map π : R → R/I satisfying
π(r) = r+ I, r ∈ R, is a nonzero homomorphism, whose kernel is equal to I. Since
π is a monomorphism iff ker(π) = {zeroR}, one can infer that π is a monomor-
phism iff I = {zeroR}, that is iff R has no proper ideals. Despite this, it was not
possible to specify such a result using just an “if and only if” formula. We parti-
tioned such lemma into two other, as shown in the Specification 39. In order to
guarantee the generality of the rings R and S, we had to let clear that these struc-
tures have no necessarily the same types. Because of this, R and S were declared
as ring variables over types T1 and T2: R: ring with one[T1,s1,p1,zero1,one1]

and S:ring[T2,s2,p2,zero2] (see Lemma mono no prop id). In order to prove
necessity of the Lemma 4 (Lemma no prop id mono), one has to provide an in-
stantiation for the homomorphism π, whose co-domain is S = R/I, but if the type
of the ring S were assigned as T2 such instantiation would not be possible since
the type of ring R/I is coset(R,I). Since the sufficiency of Lemma 4 does not
depend of the canonical homomorphism π, it is not necessary to restrict the type
of the ring S as shown in the Lemma mono no prop id.

Specification 39 Specification that differs of its enunciated in algebra textbooks

no_prop_id_mono: LEMMA
(FORALL(R:ring_with_one[T1,s1,p1,zero1,one1],I: ideal[T1,s1,p1,zero 1](R)):
(FORALL(S:ring[coset[T1,s1](R,I),add[T1,s1](R,I),product(R,I),I],
phi:R_homomorphism[T1,s1,p1,zero1,coset[T1,s1](R,I),add[T1,s1](R,I),

product(R,I),I](R,S)):
NOT zero_homomorphism ?[T1,s1,p1,zero1,coset[T1,s1](R,I),add[T1,s1](R,I),

product(R,I),I](R,S)(phi) IMPLIES
R_monomorphism ?[T1,s1,p1,zero1,coset[T1,s1](R,I),add[T1,s1](R,I),

product(R,I),I](R,S)(phi)) IMPLIES
I = R OR I = singleton(zero 1))

mono_no_prop_id: LEMMA
(FORALL(R: ring_with_one[T1,s1,p1,zero1,one1],S:ring[T2,s2,p2,zero2],
phi:R_homomorphism[T1,s1,p1,zero1,T2,s2,p2,zero 2](R, S)):
(FORALL(I: ideal[T1,s1,p1,zero 1](R)):(I = R OR I = singleton(zero 1)))
IMPLIES (NOT zero_homomorphism ?(R,S)(phi) IMPLIES
R_monomorphism ?(R,S)(phi)))

9 Conclusions and future work

This paper presents complete formalizations in PVS of the three isomorphism the-
orems for rings, central results about principal, prime and maximal ideals, speci-
fications of the main structures and concepts, and formalizations of all properties
required to obtain such formal proofs. The preliminary concepts were specified
following a general approach that allows their use in contexts involving structures
other than rings such as monoids, semigroups, groups, and fields. Furthermore,
to illustrate the advantages and possible applications of such a general algebraic
approach, an elaborated and complete formalization of the Chinese Remainder
Theorem for rings was developed. Also, it was discussed how a full formalization
of the Chinese Remainder Theorem for integers was obtained as a corollary of the
general algebraic version of CRT for rings. For doing that, additional formaliza-
tions of properties of the rings Z and Zn were necessary.

From a total of 1356 formulas formalized in theory algebra, the development
reported in this paper (rings theory) consists of 583 formalized formulas (lem-
mas and theorems), from which CRT for rings includes 315 formulas, counting
all its dependencies. The amount of formalized results indicates this formalization
contributed significantly to the enrichment of theory algebra.

Ring theory has many relevant applications in fields such as coding theory,
segmentation of digital images, cryptography, among others. In this sense, the
rings PVS theory development conforms a robust basis towards constructing more
elaborated theories involving rings, their properties, and applications.

The formalization of factorization in commutative rings with identity is an
interesting future work. Since the notion of divisibility and prime elements can



32 Thaynara Arielly de Lima et al.

be defined in commutative ring with identity, a formalization of an analog of the
Fundamental Theorem of Arithmetic for principal ideal domains would be of great
relevance. Furthermore, developments of sub-theories about rings of polynomials,
including notions of factorization and a division algorithm in such rings and the
formalization of Eisenstein’s irreducibility criterion are important landmarks that
might enrich the theory rings. Another interesting formalization would be the
concept of the ring of fractions defined from an equivalence relation over a com-
mutative ring R and a multiplicative subset S of R. This notion allows defining
localization of a ring and developing formalizations of concrete structures like the
ring Zpn , for p a prime number, as an example of local rings.

